

MATH 149

LABORATORY ASSIGNMENT 3

DERIVATIVES AND PROPERTIES OF GRAPHS

The important characteristics of the graph of a function $f(x)$ can be established by studying its first and second derivatives. These characteristics include the location of any local maxima, local minima, and points of inflection, and intervals in which the graph is increasing or decreasing, or is concave upward or concave downward.

As an example we will study the function $f(x) = x^2 + 3\sin(2x)$, $-\pi \leq x \leq \pi$.


```
> restart;
```

```
> f:=x->x^2+3*sin(2*x);
```

$$f := x \rightarrow x^2 + 3 \sin(2x)$$

First we plot the graph of the function:

```
> plot(f,-Pi..Pi);
```




There appear to be two local minima, a local maximum close to $x = 1$, and possibly another local maximum near $x = -3$. At points where there is a local maximum or minimum the derivative is 0. We next compute the derivative and draw the graphs of $f(x)$ and $f'(x)$ on the same set of axes:

```
> D(f);
```

$$x \rightarrow 2x + 6 \cos(2x)$$

```
> plot([f,D(f)],-Pi..Pi,color=[red,green]);
```



To find the exact locations of the local maxima and minima we solve the equation $f'(x) = 0$:

```
> solve(D(f)(x)=0,x);
```

$$\frac{1}{2} \text{RootOf}(_Z + 6 \cos(_Z))$$

Apparently, MAPLE does not know a general solution, so we will find the solutions using the "fsolve" command.

```
> X[1]:=fsolve(D(f)(x)=0,x=-1..0);X[2]:=fsolve(D(f)(x)=0,x=0..1);X[3]
:=fsolve(D(f)(x)=0,x=1..2);
```


$$X_1 := -0.6723755227$$

$$X_2 := 0.9457599482$$

$$X_3 := 1.992913103$$

To see if there is another solution near -3, we zoom in on the graph of $f'(x)$:

```
> plot(D(f),-Pi..-2.8,-1..1,color=blue);
```



Since the graph of $f'(x)$ does not touch the X-axis there is not an additional solution.

Note that, from the graph of $f'(x)$, $f'(x)$ is negative in the intervals (π, X_1) and (X_2, X_3) , and thus $f(x)$ is decreasing on these intervals. $f'(x)$ is positive in (X_1, X_2) and (X_3, π) , and therefore $f(x)$ is increasing there.


We now apply the First Derivative Test. Since $f'(x)$ is negative to the left of X_1 and positive to the right, there is a local minimum at X_1 ; similarly, there is a local minimum at X_3 . Since $f'(x)$ is positive to the left of X_2 and negative to the right, there is a local maximum at X_2 .

The second derivative is used to find intervals of concavity and points of inflection. We will compute $f''(x)$ and plot it and $f(x)$ on the same set of axes.

```
> D(D(f));
```

$$x \rightarrow 2 - 12 \sin(2x)$$

```
> plot([f,D(D(f))],-Pi..Pi,color=[red,blue]);
```



Points of inflection occur at points where $f''(x) = 0$ and the second derivative changes sign. The graph is concave down on intervals where $f''(x) < 0$ and concave upward when $f''(x) > 0$; thus the points of inflection are the points where the concavity changes. Here there are four such points:

```
> Z[1]:=fsolve(D(D(f))(x)=0,x=-Pi..-3);Z[2]:=fsolve(D(D(f))(x)=0,x=-2..-1);Z[3]:=fsolve(D(D(f))(x)=0,x=0..1);Z[4]:=fsolve(D(D(f))(x)=0,x=1..2);
```

$$Z_1 := -3.057868614$$

$$Z_2 := -1.654520366$$

$$Z_3 := 0.08372403961$$

$$Z_4 := 1.487072287$$

```
>
```

The graph of $f(x)$ is concave up for x in $(-\pi, Z_1)$, (Z_2, Z_3) , and (Z_4, π) . It is concave down in (Z_1, Z_2) and (Z_3, Z_4) .